# Maintainer: Maxim Baz <archlinux at maximbaz dot com>

pkgname=kak-lsp
pkgver=15.0.0
pkgrel=1
pkgdesc='Kakoune Language Server Protocol Client'
arch=('loong64' 'x86_64')
url='https://github.com/kak-lsp/kak-lsp'
license=('custom:unlicense')
depends=('kakoune')
makedepends=('rust')
optdepends=(
    'bash-language-server: language server for Bash'
    'gopls: language server for Go'
    'haskell-language-server: language server for Haskell'
    'lua-language-server: language server for Lua'
    'python-lsp-server: language server for Python'
    'rust-analyzer: language server for Rust'
)
source=("${pkgname}-${pkgver}.tar.gz::https://github.com/ul/${pkgname}/archive/v${pkgver}.tar.gz")
b2sums=('7ba882670ec08c1d2d2647f959d51d2ca9b517705a4e66322e207249066f8058d3ddcab1f5e0983323c69cdaaa60b251aa3e306cdd7513bd32dda6b96b705b9a')

build() {
    cd "${pkgname}-${pkgver}"
    make
}

package() {
    cd "${pkgname}-${pkgver}"
    make DESTDIR="${pkgdir}" install
}

# vim:set ts=4 sw=4 et:
